
1 - The Wall



Programs So Far

Generally, in 150, we have been interested in writing functions which run to
completion and compute some value along the way.

Such programs are often mathematical in nature–they only interface with the outside
world insofar as they are asked to compute some value, but the program execution
otherwise does not really care about anything having to do with the outside world1.

These are nice, but only form a subset of interesting programs. There are many
programs which are tailor-made to deal with humans!

1Not too dissimilar from some computer scientists.

Interactive Programs

Such programs are often interactive–they entail taking in some input, computing
some results by talking to other programs or servers, and require interactivity to be
preserved while potentially doing long computations in the background.

Writing such interactive programs requires the ability to do asynchronous
programming—programs which can run multiple things over a period of time.

Def A concurrent program is one which can run multiple different operations in an
interleaved fashion over a period of time.

A Program You Cannot Write (Yet)

We want to write a REPL which only has one command, ping , which causes the
REPL to return pong .

This command will also take in an optional argument, which is how many seconds
to wait before returning pong!

repl > help
commands: ping [seconds] | help | quit
repl > ping
pong! (waited 0 seconds)
repl > ping 3
pong! (waited 3 seconds)
repl > quit

The Obvious Implementation

datatype command = Help | Ping of int

fun ping n =
( wait n;

print ("pong! (waited " ^ Int.toString n ^ " seconds)")
)

fun repl () =
( print "repl > ";

case parse (input_line ()) of
Help => (print helpText; repl ())

| Ping n => (ping n; repl ())
)

For highlighted code, let’s assume that these functions exist for the moment and not
worry too much about how they work.

What Just Happened?

Makes sense. We notice something interesting, though:

repl > ping 3

The repl > prompt isn’t shown until 3 seconds later, when the ping 3 command
finishes!

This means we can’t possibly run help or even another ping command until the
first one finishes. In other words, commands in our REPL are blocking–they prevent
us from doing anything else until they finish.

Where the Blocking Lives

datatype command = Help | Ping of int

fun ping n =
( wait n;

print ("pong! (waited " ^ Int.toString n ^ " seconds)")
)

fun repl () =
( print "repl > ";

case parse (input_line ()) of
Help => (print helpText; repl ())

| Ping n => (ping n; repl ())
)

In particular, both of the above highlighted expressions are blocking—they will stall
the entire program until the world answers.

The Real Problem Is Logical, Not Mechanical

Ultimately, the problem originates from the fact that SML is run-to-completion. It
is only allowed to run one thing (function) at a time, and it cannot start progressing
on another until the first one finishes.

In other words, our call to ping : int -> unit is a monolith — it returns a unit
after a certain amount of time, and nothing is allowed to happen until it finishes.

| Ping n => (ping n; repl ())

In this case, our ping call is the problem. It blocks the entire loop, when in reality
we want to be able to proceed to the repl () call, and then in n seconds somehow
come back and do the work we need.

The Key Revelation: To-Do Lists

Our key revelation is this:

Key Asynchronous programs are synchronous programs which have to-do lists.

This means we can solve the problem by simply adding a to-do list to the program.
Our idea will be to make it so that ping can yield immediately, but also write down
somewhere what to do later, and fulfill it at a later time.

A First Attempt: Total Cooperation

Let’s build the dumbest thing that could possibly work. The rule: never commit to
waiting for anything2. Instead: check for work, nap briefly, check again, forever3.

First, ping stops performing and starts scheduling:

val pending : (Time.time * int) list ref = ref []

fun ping n =
pending := (now () + n, n) :: !pending

ping is now instant — it does nothing except write a deadline on the to-do list. The
actual pong-printing must happen somewhere else, later.

2This is not necessarily dating advice.
3This is what a job is like.

A First Attempt: Total Cooperation

Somebody has to read the list:
fun do_pongs () =

let
(* split into the ones past due and not *)
val (due , rest) =

List.partition (fn (t, _) => t <= now ()) (! pending)
in

pending := rest;
List.app (fn (_, n) =>

print ("pong! (waited " ^ Int.toString n ^ " seconds)")
) due

end

Nothing clever: split off everything whose deadline has passed, and do it. Note
nobody “fires” a timer — time never calls us; we check the time.

A First Attempt: Total Cooperation

fun tick () =
( do_pongs ();

if input_ready ()
then case parse (input_line ()) of

Help => print helpText
| Ping n => ping n

else ();
wait 1;
tick () )

One new assumed ability: input_ready asks whether a line is ready without
committing to read it. Input is never awaited — only checked.

A First Attempt: It Works!

repl > ping 3
repl > ping 1
pong! (waited 1 seconds)
pong! (waited 3 seconds)
repl > quit

This actually totally works. We can overlap pings, our prompt never dies, and
pongs arrive in the proper order. This is a concurrent program in plain, sequential
SML.

However, everything happens on the second scale. If we type help at t = 0.4, we
get answered at t = 1.0. Not to mention, the program keeps spinning forever.

A First Attempt: The Bill

• Lag is quantized to the tick. Everything is answered up to a second late. A
smaller tick means that the CPU spins, and a larger tick makes lag grow.

• tick is a god-function. It must enumerate every concern in the program; each
new feature means editing it.

• 86,400 wakeups a day4, mostly to do nothing.

And one deeper limit: our list can only remember pongs. We have no way of
deferring arbitrary work, like printing something different or printing two things in
sequence.

Here’s an idea: what if we were able to defer arbitrary work to be done later? What
does this sound like?

4Not to mention 525,600 * 60 in a year!

Rediscovering CPS

We have rediscovered the need for continuation-passing style. Before, we used
them only to write code in a different light–now, we use them to write code which is
fundamentally "pausable" at certain points.

Before, we assumed only the existence of wait : int -> unit . Now, we assume
the existence of its CPS cousin, waitThen: int -> (unit -> unit) -> unit ,
which accepts a continuation to run sometime after the wait is complete.

Importantly, waitThen yields immediately, but we will require some other entity to
run the continuation for us, sometime after — an entity much like our tick loop, just
with a smarter nap.

ping, with waitThen

datatype command = Help | Ping of int

fun ping n =
waitThen n (fn () =>

print ("pong! (waited " ^ Int.toString n ^ " seconds)")
)

fun repl () =
( print "repl > ";

case parse (input_line ()) of
Help => (print helpText; repl ())

| Ping n => (ping n; repl ())
)

One More Bug

We have one more problem, which is that there is still one more bug left to squash!

repl > ping 3
repl >

After we type ping 3, the prompt returns instantly, but we still do not see the pong!

The problem is that the call to input_line is still blocking, meaning that our
continuation that prints the pong can’t go while our repl is waiting. We need a
second ability: the ability to do some other stuff while waiting for someone to type
something.

The Same Move for Input

We will apply a similar technique–instead of blocking on the result of input_line ,
we will hand it a function to call once that line arrives, and then immediately return.

We assume the existence of inputLineThen : (string -> unit) -> unit ,
which accepts a continuation to run when the line (optionally) is available.

A Subtle Issue

fun repl () =
( print "repl > ";

inputLineThen (fn line =>
case parse line of

Help => print helpText
| Ping n => ping n

);
repl ()

)

Only, be careful here. There’s a subtle issue.

Spot the Difference

fun repl () =
( print "repl > ";

inputLineThen (fn line =>
case parse line of

Help => print helpText
| Ping n => ping n
);
repl ()

)

fun repl () =
( print "repl > ";

inputLineThen (fn line =>
(case parse line of

Help => print helpText
| Ping n => ping n);
repl ()

)
)

What is the difference between these two implementations?

In a world where we assume inputLineThen to return immediately, the first
actually loops forever and spam prints "repl >" forever.

The Fix

The lesson, which we just learned the hard way: after a registration, the only
“later” is inside the callback. So the recursive call must move inside:

fun repl () =
( print "repl > ";

inputLineThen (fn line =>
( case parse line of

Help => print helpText
| Ping n => ping n

; repl () ))
)

Each arrival of a line handles one command and re-arms the next read. The REPL’s
“loop” is no longer a loop we run — it’s a chain of registrations, each planting the
next.

2 - Callbacks and Promises



Two Suppositions

Everything we’ve developed so far is based on two suppositions:

First, we assume that we have the ability to wait and accept input "in the
background", in a way that will not block our REPL. This is exhibited by the following
functions:

val waitThen : int -> (unit -> unit) -> unit
val inputLineThen : (string -> unit) -> unit

We also assume the ability of some master process which decides when to run our
continuations.

The former is the ability to run non-blocking operations, and the latter is known as
the event loop.

The Two Registries

(* two registries: our timers , and one pending read *)
val timers : (Time.time * (unit -> unit)) list ref = ref []
val read_job : (string -> unit) option ref = ref NONE

fun waitThen time f = timers := (now () + time , f) :: !timers
fun inputLineThen k = read_job := SOME k

Our two “magic” primitives are one line each: they perform nothing, they file
paperwork. All the actual running happens in one place:

The Event Loop

fun loop () =
case (try_input (), !read_job , !timers) of

(SOME line , SOME k, _) =>
(read_job := NONE; k line; loop ())

| (_, _, (t, f) :: rest) =>
if now () >= t
then (timers := rest; f (); loop ())
else (nap (); loop ())

| (_, NONE , []) => ()
| _ => (nap (); loop ())

Crucially, nap is not the tick loop’s blind one-second snooze: it blocks until input is
ready or the earliest deadline passes, whichever first. This is the promised
smarter nap — we wake exactly when there is something to do. No quantized lag,
no wakeups for nothing.

The REPL, Shattered

fun ping n =
waitThen n (fn () =>

print ("pong! (waited " ^ Int.toString n ^ " seconds)")
)

fun repl_step line =
( case parse line of

Help => (print helpText)
| Ping n => (ping n);
ask ()

)

and ask () =
( print "repl > ";

inputLineThen (fn line => repl_step line)
)

One Registry per Operation?

This works, we need one piece of mutable state for every kind of operation we
anticipate needing to fulfill later (a list of timers for each waitThen , and a single
read job for every pending inputLineThen).

If we wanted to add more (HTTP requests, DB reads, etc) we would need to add
more state and more kinds of callbacks, and more cases into our event loop.

Is there a generic way we can handle this?

A New Program: Page Similarity

Let’s dispense with the REPL for a moment. Suppose we’re interested in a very
simple sequence of operations: we want to read two names, fetch their Wikipedia
pages, and then compute a similarity score between the two.

val name1 = read_line ()
val name2 = read_line ()
val page1 = fetch_page name1
val page2 = fetch_page name2
val score = compute_similarity page1 page2

This is a totally blocking program, as it currently stands. Let’s shatter it into its
constituent parts, in the hopes of letting other parts run between steps.

Callback Hell

The easy way to do this is to just arm each operation with a continuation, per usual.
We’ll talk about what to do with the continuation later.

read_line (fn name1 =>
read_line (fn name2 =>

fetch_page name1 (fn page1 =>
fetch_page name2 (fn page2 =>

compute_similarity page1 page2 score
)

)
)

)

This is commonly what is known in JavaScript as callback hell.

Two Problems

The code is gross. The problem is similar to what we saw in the CPS lecture–we
induce one level of indentation per continuation.

Additionally, we have to prescribe the order in which the continuations run. In
reality, it’s not so important that we literally fetch the first name’s page before the
second name’s page–we should actually be able to do them at the same time.

We want a more general mechanism for this idea of having pending work to do with
a value that we don’t yet have. This is the idea of a promise.

Promises

Def A promise is a value which represents the result of an asynchronous operation
that we do not yet have.

In other words, a t Promise.t is a promise which will eventually contain a value
of type t.

This is the general mechanism we were looking for. We previously changed our
functions from blocking ones to ones that returned immediately, but accepted
continuations telling us what to do once we had the value. This is encoded via:

(* a direct function *)
’a -> ’b

(* an asynchronous function *)
’a -> ’b Promise.t

thenDo

The point is that once we have a promise, we are allowed to register a continuation
on it, which is automatically queued to run once the promise is fulfilled.
So, instead of the callback hell we had before, we can now define an infix operator
thenDo :

thenDo : ’a Promise.t * (’a -> ’b Promise.t) -> ’b Promise.t

(read_line ()) thenDo (fn name1 =>
(read_line ()) thenDo (fn name2 =>
(fetch_page name1) thenDo (fn page1 =>
(fetch_page name2) thenDo (fn page2 =>
compute_similarity page1 page2
))))

This isn’t much better, actually. But, we can improve things:

both

read_line () thenDo (fn name1 =>
read_line () thenDo (fn name2 =>

both (fetch_page name1 , fetch_page name2)
thenDo (fn (page1 , page2) =>

compute_similarity page1 page2)))

We can use a primitive called
both : ’a Promise.t * ’b Promise.t -> (’a * ’b) Promise.t to
combine two promises into one. Essentially, it becomes one promise which is
fulfilled only when both of the original promises are fulfilled.

Importantly, this does not specify which fetch finishes first! Both are started
together, and we only compute the score once both pages have arrived. (The two
reads stay sequential on purpose — there’s only one keyboard.)

The Signature

signature PROMISE =
sig

type ’a t
type ’a promise = ’a t

val new : unit -> ’a promise
val resolve : ’a promise * ’a -> unit
val upon : ’a promise * (’a -> unit) -> unit

val thenDo : ’a promise * (’a -> ’b promise) -> ’b promise
val both : ’a promise * ’b promise -> (’a * ’b) promise

end

The Promise REPL

Let’s first see what we’re building toward. Assume our two primitives now return
promises instead of taking callbacks:

val sleep : int -> unit promise
val read_line : unit -> string promise

And one more operation: upon (p, k) registers a plain observer — run k on the
value once it exists. (Like thenDo , except k returns nothing, and no new promise is
created.)

The contract underneath: when a promise resolves, everything waiting on it drops
into the loop’s inbox of ready work.

Before we see how to build this, let’s see how it would be used.

The REPL, Rewritten

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited " ^ Int.toString n ^ " seconds)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText; repl ())
| Ping n => (ping n; repl ()))

)

This is a similar shape to the first repl we wrote, but every wait is now a
registration, the recursion lives inside the continuation, and nothing blocks.5

5From here on, the runtime owns the event loop: every program implicitly ends by entering it — just
as in JavaScript, where the host runs your script to completion, then loops.

3 - A Promise Trace



Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: repl () runs to completion; the run-
time then (implicitly) enters its event loop.

waiting on the world
(nothing)

inbox (todo)
(empty)

terminal

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: repl runs, start to finish, uninter-
rupted.

waiting on the world
(nothing)

inbox (todo)
(empty)

terminal
repl>

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: read_line parks “resolve P1” with
the reader; upon parks the handler on P1.
repl RETURNS.

waiting on the world
reader → P1 [line handler]

inbox (todo)
(empty)

terminal
repl>

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: loop: inbox empty, no timers due —
sleep until input arrives. (Nothing of ours is
running.)

waiting on the world
reader → P1 [line handler]

inbox (todo)
(empty)

terminal
repl>

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: WORLD: user types “ping 2”. Loop
wakes; reader fires resolve P1; P1’s waiters
move to the inbox.

waiting on the world
(nothing)

inbox (todo)
handler (“ping 2”)

terminal
repl> ping 2

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: loop runs the handler: parse the
line...

waiting on the world
(nothing)

inbox (todo)
(empty)

terminal
repl> ping 2

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: ...it’s Ping 2, so call ping 2.

waiting on the world
(nothing)

inbox (todo)
(empty)

terminal
repl> ping 2

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: sleep 2 files a timer that will resolve
P2; upon parks the pong printer on P2.

waiting on the world
timer +2s → P2 [print pong]

inbox (todo)
(empty)

terminal
repl> ping 2

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: ...then repl () re-arms: new prompt,
new read (P3). The handler RETURNS; the
step is over.

waiting on the world
timer +2s → P2 [print pong]
reader → P3 [line handler]

inbox (todo)
(empty)

terminal
repl> ping 2
repl>

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: loop: sleep until input OR the +2s
deadline, whichever first. (Nothing of ours is
running.)

waiting on the world
timer +2s → P2 [print pong]
reader → P3 [line handler]

inbox (todo)
(empty)

terminal
repl> ping 2
repl>

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: WORLD: 2 seconds pass. Timer
due; resolve P2 fires; P2’s waiter moves to
the inbox.

waiting on the world
reader → P3 [line handler]

inbox (todo)
print pong

terminal
repl> ping 2
repl>

Trace: The Promise REPL

fun ping n =
upon (sleep n, fn () =>

print ("pong! (waited ...)"))

fun repl () =
( print "repl > ";

upon (read_line (), fn line =>
case parse line of

Help => (print helpText;
repl ())

| Ping n => (ping n;
repl ())) )

val () = repl ()

system: loop runs the pong printer. Then: in-
box empty, one reader parked — sleep until
input. The REPL lives on.

waiting on the world
reader → P3 [line handler]

inbox (todo)
(empty)

terminal
repl> ping 2
repl> pong! (waited 2 seconds)

4 - Building Promises



Three Drafts of a Cache, Three Diseases

Before we build promises, let’s see what can go wrong. Raw callbacks come with no
guarantees: nothing checks how a function treats the continuation it was handed.

Here’s an innocent task: fetching a page is slow, so cache it — call k straight from
the cache when we can.

We’ll write it three times, and each time see that it breaks a different unwritten rule.

Draft 1: The Dropped Continuation

fun fetch_cached name k =
if valid name
then fetch_page name (fn page =>

(insert (name , page); (* stash for next time *)
k page))

else print "bad name!"

On a bad name, k is never called. The caller doesn’t crash — it waits forever,
silently. Everything that was going to happen next was inside k, and nobody ran it.

Why it’s so nasty: the failure is an absence. No error, no stack trace, no line to point
at — the program didn’t die, it just stopped having a future.

Draft 2: The Double Call

fun fetch_cached name k =
( case lookup name of

SOME page => k page (* meant to stop here! *)
| NONE => ()

; fetch_page name (fn page =>
(insert (name , page); k page)) )

On a cache hit, k runs now — and again when the fetch lands. But k is the rest of
the program: two pongs, two prompts, two of everything downstream.

Nothing errors at the call site; the wreckage appears later, elsewhere. (In
JavaScript this is the famous “forgot the return before the callback” — arguably
the most-committed callback bug in history.)

Brandon Wu Asynchronous Programming 16 July 2026 55 / 71



Draft 3: Sometimes Now, Sometimes Later

Third try — and this one looks correct:

fun fetch_cached name k =
case lookup name of

SOME page => k page (* HIT: k runs NOW *)
| NONE => fetch_page name (fn page =>

(insert (name , page); k page)) (* MISS: k runs LATER *)

Now look at it from a caller’s seat:

( fetch_cached name (fn page => print page)
; print "done!" )

Which prints first, the page or the done? On a hit, print page runs first, but on a
miss, it runs second. It’s all dependent on the cache’s mood.

...and One Disease of Our Own

The library isn’t the only one who can sin — callers can too. Remember our
spam-printing repl? Its bug was the code placed after the registration — the
repl () call ran immediately. Here is that same disease, minimized:

( inputLineThen (fn line => print line)
; print "done!" )

done! prints first. Registration returns instantly; the code after it runs now, not after
the line arrives.

Key After a registration, the only “later” is inside the callback.

The Tally

• Dropped: some branch never calls k — the workflow waits forever, silently.
• Doubled: some branch calls k twice — the rest of the program happens twice.
• Sometimes-now: k runs before the call returns on some inputs — order at the

call site depends on hidden state.
• Premature after: code placed after a registration runs before the result exists.

None of these require carelessness — every branch of every callback-taking
function must maintain “exactly once, always later” by hand, forever, in every
library you depend on.

Hold that phrase. We’re about to make it a datatype.

Two Kinds of Queue

Our conceptual view of our computational model is that we have two kinds of
queues:

• The run queue (one, owned by the event loop): contains jobs, which are ready
to run anytime.

• The wait queues (one inside every pending promise): continuations linked to
each promise, which are ready to receive the promise’s future value. These
cannot run yet, as we do not yet have the value

Essentially, a promise contains potential run queue functions, which are only
promoted once the promise is fulfilled.

Brandon Wu Asynchronous Programming 16 July 2026 59 / 71



Implementing the Run Queue

Here is how we implement the run queue:

val todo : (unit -> unit) list ref = ref []

fun defer f = todo := !todo @ [f]

This is the generality we’ve been chasing since the tick loop, finally delivered:
deferred computation with no strings attached. A todo entry is simply code to
run, later, no qualifications.
The loop drains it first, before anything else:

fun loop () =
case !todo of

f :: fs => (todo := fs; f (); loop ())
| [] => (* input , timers , nap: as before *) ...

This is the first and last change the loop will ever need.

Implementing Promises

Promises sound like magic, but we’ve essentially already built them.

val read_job : (string -> unit) option ref = ref NONE

A slot where a continuation waits for a string is essentially a promise with two
limitations: it forgets the value once delivered, and it holds only one waiter.

datatype ’a state =
Pending of (’a -> unit) list

| Resolved of ’a

type ’a promise = ’a state ref

upon and resolve

fun upon (p, k) =
case !p of

Resolved v => defer (fn () => k v) (* 1 *)
| Pending ks => p := Pending (k :: ks)

fun resolve (p, v) =
case !p of

Resolved _ => raise Fail "resolved twice!" (* 2 *)
| Pending ks =>

( p := Resolved v
; List.app (fn k => defer (fn () => k v)) ks )

This construction carefully avoids the bugs we previously outlined. 1 ensures we
always defer our continuation, even if the answer is present, and 2 ensures
resolving twice raises so we do not double-call.

Wrapping the Primitives

Where do promises come from? We wrap the primitives we already have:

(* an empty promise is an empty queue of continuations *)
fun new () = ref (Pending [])

fun sleep n =
let val p = new ()
in waitThen n (fn () => resolve (p, ())); p end

fun read_line () =
let val p = new ()
in inputLineThen (fn line => resolve (p, line)); p end

Notice what did not change: waitThen , inputLineThen , the timers list, the read
job. Promises replaced nothing — they organize what we had. A promise is a
parking spot; resolve is a courier to the same old loop.

What Changed in the Event Loop?

We saw earlier that we changed the inbox to make it more general.

The real advantage we have gained here is that promises both compose nicely and
also prevent us from committing a class of bugs (the ones we saw earlier).

And nothing about waiting changed: the loop still sleeps until input or the next
deadline — promises are a library above the loop, not machinery inside it. Any
future kind of operation (HTTP requests, database reads, ...) is one sleep-style
wrapper away; the loop gains no new cases, ever.

5 - Asynchronous Code and TypeScript



Asynchronous TypeScript

Today’s lecture involved trying to phrase asynchronous code in a
run-to-completion single-threaded program, with access to higher-order
functions.

We’re lucky that we weren’t the first to have to do this–JavaScript was one of the
first languages to tread this path, long ago!

Everything we implemented today has a direct analogue somewhere in the history
of JavaScript.

async/await, Compiled

It turns out that frontend development just ends up being functional programming
all along.

2009 callbacks (Node.js) = CPS (L11)
2011 async.js combinators = HOFs (L10)
2015 promises (ES6) = CPS + single-assignment ref (L19)
2017 async/await (ES2017) = the CPS transform, by compiler (L11)

The very themes you have been learning in this course are the same as powered
the entire JavaScript and TypeScript ecosystems, essentially.

Brandon Wu Asynchronous Programming 16 July 2026 67 / 71



Asynchronous TypeScript

TypeScript has a lot of nice abstractions for dealing with asynchronous code. You
can write:

const result = await fetch("https :// example.com");
const data = await result.json();
console.log(data);

This code just compiles down to our promise-based equivalent!

fetch "https :// example.com" thenDo (fn result =>
jsonify result thenDo (fn data =>

print data
)

)

The Reveal

All the niceties of asynchronous programming in Typescript, we saw today.

TypeScript Today
Promise<A> A Promise.t
p.then(f) thenMap or thenDo
await p p thenDo (fn x => ...) (compiled!)
Promise.all([p, q]) both (p, q)
async (a: A) => B A -> B Promise.t

What We Hid (On Purpose)

The actual system layer underneath consists of threads, processes, sockets,
syscalls, none of which we needed to get into to reason about the application layer
correctly. However, there are a few things we glossed over that could improve the
design:

• Errors: our promises only succeed. Real ones carry a rejection channel
(Rejected of exn), so failures flow down chains to one handler — JS’s
.catch .

• Type safety: a proper promise library would separate the type of the promise
from its resolver, so readers can’t resolve.

• Compute-heavy jobs: all our work waits. A job that computes for 3s freezes
the loop; fixing that needs actual concurrency — threads, processes, workers.

• Cancellation: nothing can un-register a pending workflow. Done right,
suspensions become values a runner can decline to continue.

• Bytes, not lines: input arrives in chunks, and “readable” doesn’t mean a whole
line is there — real hosts keep buffers.

The Whole Lecture, in Five Lines

• Asynchronous programs are synchronous programs with to-do lists. A main
loop, some registries, and some well-informed naps.

• After a registration, the only “later” is inside the callback — code placed after it
runs now.

• A promise is the answer as a first-class value: a list of parties waiting, or the
answer itself, which is delivered exactly once by construction.

• await is a compiled promise chain. It’s not magic, it’s just CPS where you
don’t have to write the continuations.

• Asynchronous programming boils down to functional programming: first-class
functions to write the future down, CPS to shatter it, and a ref to remember the
details.

